

Monte Carlo Numerical Evaluation of a Definite Integral - f-average Method

Created using Maple 14.01

Jake Bobowski


```
> restart;
with(stats) :
with(plots) :
with(Statistics) :
with(StringTools) :
FormatTime("%m-%d-%Y, %H:%M");
"03-21-2013, 21:23" (1)
```

In all Monte Carlo simulations it is necessary to generate random or pseudo-random numbers. The following statement will generate a random number drawn from a uniform distribution between 0 and 1.

```
> X := x → stats[random, uniform[0, 1]](1) :
X();
0.3957188605 (2)
```

This tutorial will attempt to numerically evaluate an integral for which the exact solution is easily obtained. This approach has been taken purposely so that we can confirm that our numerical techniques are reliable. The function that we will integrate is a simple polynomial. Below the function is plotted on the interval $x = 0..1$ and the exact value of the integral is evaluated over the same interval.

```
> fcn :=  $\frac{1}{27} (-65536 \cdot x^8 + 262144 \cdot x^7 - 409600 \cdot x^6 + 311296 \cdot x^5 - 114688 \cdot x^4 + 16384 \cdot x^3)$  :
plot(fcn, x = 0 .. 1, axes = boxed, view = [0 .. 1, 0 .. 1.1], labels = [typeset("x"),
typeset("f(x)"), labeldirections = ["horizontal", "vertical"], symbol = circle, symbolsize
= 20, thickness = 2, tickmarks = [8, 8], colour = blue, axesfont = [Times, 12], labelfont
= [Times, 14], axis = [gridlines = [thickness = 2]]);
IntExact := evalf(int(fcn, x = 0 .. 1));
```



IntExact := 0.4815990594

(3)

The f-average method requires that we repeatedly generate x_i random numbers (n times) uniformly distributed between the integration interval (0 to 1 in this example). For each x_i we calculate the corresponding $f(x_i)$ and sum all of these to find f_{total} . f_{average} is obtained from $f_{\text{average}} = f_{\text{total}}/n$. An estimate of the integral is given by the integration interval $(b-a) \cdot f_{\text{average}}$. In the plot below, we show each of the $f(x_i)$ and f_{average} which is our estimate of the integral. The square area below the red line is approximately equal to the area beneath the blue curve which is the integral that we're trying to evaluate.


```
> n := 10000 :
  xList := NULL :
  fList := NULL :
  ftot := 0 :
  for i from 1 to n do:
    x := X( ) :
    xList := xList, x :
    fList := fList, fcn :
    ftot := ftot + fcn :
  end do:
  xList := [xList] :
  fList := [fList] :
```

```

x := 'x':
favg :=  $\frac{ftot}{n}$ ;
fPlot := ScatterPlot(xList, fList, axes = boxed, view = [0 .. 1, 0 .. 1], labels = [typeset("x"),
typeset("f(x) ") ], labeldirections = ["horizontal", "vertical"], symbol = circle, symbolsize
= 15, thickness = 2, tickmarks = [8, 8], colour = blue , axesfont = [Times, 12] , labelfont
= [Times, 14], axis = [gridlines = [thickness = 2]]) :
favgPlot := plot(favg, z = 0 .. 1) :
display(fPlot, favgPlot);

```

favg := 0.4818184489



If you don't need to keep track of the individual x_i and $f(x_i)$ the calculation can be done more compactly and the loop will complete in a shorter time.

```

> n := 10000 :
ftot := 0 :
for i from 1 to n do:
  x := X() :
  ftot := ftot + fcn :
end do:
x := 'x':

```

$$IntEst := \frac{ftot}{n};$$

$$IntEst := 0.4772148481$$


(4)

Now we will numerically approximate the integral using $n = 1000$ one hundred times and plot the resulting distribution of our determination of the integral. The distribution is expected to be Gaussian. **WARNING:** Depending on the machine that your working with, this chunk of could could take some time to fully execute!

```
> FormatTime("%M:%S");
n := 10e3 :
intList := NULL :
for j from 1 to 100 do:
  ftot := 0 :
  for i from 1 to n do:
    x := X() :
    ftot := ftot + fcn :
  end do:
  intList := intList,  $\frac{ftot}{n}$  :
end do:
intList := [intList] :
FormatTime("%M:%S");
Histogram(intList, axes = boxed, view = [0.465 .. 0.495 , 0 .. 150], labels
= [typeset("Histogram"), typeset("Definite Integral")], labeldirections = ["horizontal",
"vertical"], symbol = circle, symbolsize = 20, thickness = 2, tickmarks = [8, 8], colour
= blue , axesfont = [Times, 12], labelfont = [Times, 14], axis = [gridlines = [thickness
= 2]]);
x := 'x':
```

"23:58"

"26:42"



The width of the distribution can be calculated from the standard deviation of the list of 100 approximations of the integral and is an estimate in the uncertainty of our determination of the definite integral.

```
> StandardDeviation(intList);
```

```
0.00325186606628093
```

(5)

Finally, the last thing we'll attempt to do is to understand how our uncertainty (standard deviation) depends on the n . So far, all of our calculations have used $n = 10e3$. Now we'll determine the standard deviation for values of n that range from 100 to $10e3$. Again, this block of code could take some time to complete. It took my laptop about 5 minutes...

```
> FormatTime("%M:%S");
nList := [100, 200, 500, 1000, 2000, 5000, 10e3]:
sigmaList := NULL:
for k from 1 to nops(nList) do:
    n := nList[k]:
    intList := NULL:
    for j from 1 to 100 do:
        ftot := 0:
        for i from 1 to n do:
```

```

    x := X( ) :
    ftot := ftot + fcn :
  end do:
  intList := intList,  $\frac{ftot}{n}$  :
  end do:
  intList := [intList] :
  sigmaList := sigmaList, StandardDeviation(intList) :
  print(k);
  end do:
  FormatTime("%M:%S");
  sigmaList := [sigmaList];
  x := 'x':

```

"26:42"

1
2
3
4
5
6
7

"31:45"

```

sigmaList := [0.0375580009842308, 0.0268968437069594, 0.0143433162723734,
0.0119463905767336, 0.00787880140025590, 0.00550562468588672,
0.00371406652644908 ]

```


(6)

Below we plot the uncertainty in the numerical integral estimation as a function of n (the number of trials in the Monte Carlo simulation). As expected, the uncertainty decreases as the number of trials increases. The sigma values are proportional to $1/\sqrt{n}$.

```

> ScatterPlot(nList, sigmaList, axes = boxed, view = [0 .. 11e3, 0 .. 0.05], labels = [typeset("n"),
typeset("uncertainty")], labeldirections = ["horizontal", "vertical"], symbol = circle,
symbolsize = 15, thickness = 2, tickmarks = [8, 8], colour = blue, axesfont = [Times, 12],
labelfont = [Times, 14], axis = [gridlines = [thickness = 2]]);

```



To get a better appreciation of the dependence of sigma on n , below we plot sigma as a function of $1/\sqrt{n}$ and observe the linear relationship between the two. Beautiful! All of this generated from uniformly distributed random numbers! Take a moment to reflect on what we've accomplished. We've used Monte Carlo simulations to study the behaviour of Monte Carlo simulations! The objective of a Monte Carlo calculation is always to study the characteristics of some system (often a physical system) by simulating data using random numbers.

```
> SqrtInvn := [seq(1/sqrt(nList[i]), i = 1 .. nops(nList))]:
ScatterPlot(SqrtInvn, sigmaList, axes = boxed, view = [0 .. 0.11, 0 .. 0.05], labels
= [typeset("1/sqrt(n)", typeset("uncertainty")), labeldirections = ["horizontal",
"vertical"], symbol = circle, symbolsize = 15, thickness = 2, tickmarks = [8, 8], colour
= blue, axesfont = [Times, 12], labelfont = [Times, 14], axis = [gridlines = [thickness
= 2]]);
```

